unittest
{
//	assert( [ "3", "-2 5 -1" ].parse.expand.solve == "2\n2 3\n3 3" );
	assert( [ "3", "-2 5 -1" ].parse.expand.solve == "2\n2 3\n2 3" );
	assert( [ "2", "-1 -3" ].parse.expand.solve == "1\n2 1" );
	assert( [ "5", "0 0 0 0 0" ].parse.expand.solve == "0" );
}



import std.conv;
import std.format;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( as );
}

auto solve( long[] as )
{
	// 最大値と最小値を検索
	auto ma = 0L;
	auto mi = 0L;
	foreach( i, a; as )
	{
		if( as[ ma ] < a ) ma = i;
		if( a < as[ mi ] ) mi = i;
	}
	
	// 最大値と最小値のうち、絶対値の大きいほうを使う
	string[] anss;
	if( -as[ mi ] <= as[ ma ] )
	{
		// 最大値を初期値にして累積和的に計算
		for( auto i = 1L; i < as.length; i++ )
		{
			if( as[ i - 1 ] <= as[ i ] ) continue;
			while( as[ i ] < as[ i - 1 ] )		// 最大2回足し算すれば左より右のほうが大きくなる
			{
				as[ i ] += as[ ma ];
				anss ~= format( "%s %s", 1 + ma, 1 + i );
			}
			ma = i;
		}
	}
	else
	{
		// 最小値を初期値にして累積和的に計算
		for( auto i = as.length.signed - 2; 0 <= i; i-- )
		{
			if( as[ i ] <= as[ i + 1 ] ) continue;
			while( as[ i + 1 ] < as[ i ] )		// 最大2回足し算すれば左より右のほうが大きくなる
			{
				as[ i ] += as[ mi ];
				anss ~= format( "%s %s", 1 + mi, 1 + i );
			}
			mi = i;
		}
	}
	if( anss.length == 0 ) return "0";
	else                   return format( "%s\n%s", anss.length, anss.join( "\n" ) );
}
